
Abstract


The implementation of RISC and CISC architectures determine the efficiency at which a given computer system will achieve due to the complexity and number of register transfers required to complete an operation. Additionally the architectures differ in the allocation of instruction register bits to opcodes and operands determining the maximum number of operations and memory addresses available. The purpose of this experiment was to compare the CISC and RISC architectures with respect to system pipeline efficiency and maximum clock speed. 

The RISC architecture yields the capability of faster clock speeds by reducing the worst case propagation delay in the implementation of the arithmetic logic unit. Additionally the RISC system's processor pipeline allows for considerable increases in memory data transfer through the use of concurrent data latching, and the ability to address more memory without a reduction in system capability.

Introduction


The implementation of an instruction set architecture (ISA) determines the set of procedures by which numerical operations are preformed in a given processor pipeline. The two main types of computer architecture differ in their opcode control and register transfer operations occurring during the fetch/decode/execute cycle. The architecture of a given simple computer is implemented in a register transfer language (RTL), such as VHDL, which describes the register transfers occurring during a computing cycle.

A basic computer consists of several registers, an arithmetic logic unit (ALU), memory, and a control unit state machine interconnected through an address and data bus. The control unit determines the state of an instruction in the processor pipeline. The control unit cycles between fetch decode and execute states, transitioning on the rising edge of the clock input. During the fetch state, the program counter (PC) is Loaded into the memory address register (MAR), accessing the instruction located in the memory data register (MDR). The MDR value is latched to the instruction register (IR) and the address of the operand contained in the least significant bits of the MDR is latched to the MAR, identifying the data required during the execute cycle. The PC is incremented and the state transitions to decode on the next rising clock edge.

During the decode cycle the most significant bits of the IR containing the opcode is identified and the control unit sets the state for the execute cycle. During the execute cycle, data in the accumulator (AC), and MDR is processed by the ALU, stored or loaded from memory. After the completion of the decoded operation the state returns to fetch and the next instruction is loaded.

RISC implementation yields a faster processor pipeline that is based on a limited set of opcodes designed to execute in rapid succession. The reduction of the opcode length allows more bits to be allocated for memory addressing yielding access to more memory and consequently a greater program length.

CISC implementation yields a slower processor pipeline that utilizes an extended set of opcodes designed to execute slower while computing complex, multi-step operations. While each opcode executes a more rigorous computation, allowing assembly languages to be coded with fewer instructions, the logic implementation for executing the corresponding register transfers increases execution time and requires a lower clock speed. The CISC implementation of a single instruction is often less efficient then a corresponding RISC computation requiring multiple instructions.

This experiment focused on comparing a 16 bit simple computer implemented in both reduced instruction set computing (RISC) and complex instruction set computing (CISC) computing architectures. Of the architectures provided, the "new scomp" implemented a RISC based ISA while the "old scomp" implemented a CISC based ISA. While the CISC architecture was only analyzed, the RISC implementation ISA was modified to include an extended instruction set.

Procedures


The RISC VHDL implementation of a 16 bit simple computer was programmed for a simple computation, using a memory initialization file (MIF) and was subsequently compiled. The implementation was analyzed for maximum frequency and logic cell utilization on the Altera CPLD using the MAX+ Plus II timing analyzer and compilation report. A simulation waveform was generated to verify correct accumulator values and state transitions. The previously determined maximum frequency was used to select the proper frequency division from the system clock. The configuration was tested and the clock signal was verified on an oscilloscope.

The instruction set was then expanded to include SUB, JNEG, JPOS, JZERO, OR, XOR, and ADDI instructions. The RISC implementation of these instructions utilized a 10 bit operand differentiating from the 8 bit operand CISC architecture. The 6 bit opcode was used to decode the instructions and determine the execute state of the computer.

The implementation of SUB, OR, and XOR instructions required an execution of the respective operation through the ALU and subsequent latch of ALU data to the accumulator followed by a transition to the fetch state.

The Implementation of JNEG, JPOS, JZERO, and ADDI instructions required the processing of two's compliment numbers. JNEG would assign the 10 LSB of the IR register to PC if the AC was not equal to zero and the MSB of AC was equal to one, representing a negative number in two's compliment. JPOS would assign the 10 LSB of the IR register to PC if the AC was not equal to zero and the MSB of AC was equal to zero, representing a positive number in two's compliment. JZERO would assign the 10 LSB of the IR register to PC if the AC was equal to zero.

The ADDI instruction determines if the operand stored in the 10 LSB of IR is negative or positive by evaluating the operand's MSB. If the operand was positive (MSB = 0), IR(9) & "000000" & IR(8 DOWNTO 0) was added to AC. If the operand was negative IR(9) & "111111" & IR(8 DOWNTO 0) was added to AC.

In all added instructions the RISC implementation required the use of the 10 LSB of the instruction register instead of the 8 LSB used in the CISC implementation.
Results

The maximum clock frequency of the RISC implementation was determined to be 7.9 MHz. After expansion of the instruction set the maximum frequency was reduced to 7.39 MHz corresponding to the greater worst case propagation delay through the logic arrays required to implement more complicated instructions such as ADDI. Generally a CISC implementation would include more register transfers per execution cycle and therefore have a slower clock speed. The clock input to the simple computer was set and verified to be 1 MHz (Figure 1). Although the maximum frequency was determined to be 7.39 MHz, the frequency divider provided in the Altera toolkit lacks frequency outputs between 10MHz and 1MHz. In order to remain within design capabilities a 1MHz clock rate was selected.

A MIF program was coded to execute A = (B + C) + D (Appendix A). The simulation waveform (Figure 2) shows the accumulator values as three memory locations are sequentially added to the accumulator. Values B, C, and D were chosen to be 4, 3, and 3 respectively. The simple correctly adds the memory values to the accumulator to result in am output of &H000A. The CISC architecture would require one more clock cycle to complete this operation due to the three step store process during the execute state compared to the two step process in the RISC architecture.

The RISC instruction set was then extended to include SUB, JNEG, JPOS, JZERO, OR, XOR, and ADDI instructions (Appendix B). The RISC implementation of these instructions utilized a 6 bit opcode and a 10 bit MAR. The processing of instructions, such as ADDI, requiring internal conditionals and multiple register transfers decreased the maximum clock speed of the computer implementation and increased chip resource utilization. 
Conclusions


The RISC architecture of the simple computer yields a faster processing pipeline and maximum clock frequency by simplifying the implementation of internal logic leading to a lower worst case propagation delay and requiring fewer CPLD resources. Analysis of the RISC architecture with an extended instruction set determined that multiple register transfers, internal conditionals in the execute state and memory transfer operations are the limiting factor of maximum clock speed. Similarly if a full CISC architecture had been implemented, the instruction set complexity would have used considerably more chip resources and required a lower clock speed. Additionally the RISC architecture allows for a larger program and variable utilization by extending the length of the memory address register and program counter.

The RISC architecture's logic arrangement allows for faster pipeline operation during transfers of date from system memory and latching of the memory address register. In the RISC based system the MAR is latched concurrently during the fetch cycle reducing the propagation time in the system while the CISC implementation latches the PC value into the MAR during the end of each individual execute cycle. Additionally the RISC system implements a more advanced pipeline for the STORE operation then the CISC architecture. The CISC architecture latches the memory write (MW) signal during the execute state requiring three clock cycles to execute a memory store operation to ensure proper stabilization of the MW signal. The RISC architecture latched the MW signal in the decode state allowing stabilization during the execute phase and requiring one less clock cycle to complete.

Analysis of RISC and CISC system architectures determined that while both configurations implement 16 bit accumulators, instruction registers, and memory data registers, allowing for 16 bit two's compliment data processing, the difference in opcode and operand length allowed the RISC architecture to perform operations faster and access more memory then the corresponding CISC system. The RISC implementation on a 16 bit system allocates 6 bits to the opcode and 10 bits to the memory address register. While this limits the RISC architecture to 64 opcodes, it allows access to 1024 memory locations for data and program space allocated to the MIF file. The CISC implementation 8 bits to the opcode and 8 bits to the memory address register dividing the addressing capabilities equally between 256 opcodes and 256 memory locations.


In most system configurations the ability to decode 64 opcodes exceeds the computational requirements of a given implementation and allows the capability to address more memory. Additionally a RISC architecture allows a simpler logic implementation and the use of fewer chip resources. The RISC system would be preferable due to it's faster clock and execution speed, as well as the ability to allocate more program and data memory. 
Appendix A: MIF Program
-- Altera Memory Initialization File (MIF)

DEPTH = 1024

WIDTH = 16

ADDRESS_RADIX = HEX;

DATA_RADIX = HEX;

CONTENT

  BEGIN

    [000..3FF] : 0000;  -- Default to NOP

           000 : 0411;  -- Start:  LOAD    B       ;Load value stored in B, MEM(11)

           001 : 0C12;  --         ADD     C       ;Add value stored in C, MEM(12)

           002 : 0C13;  --         ADD     D       ;Add value stored in D, MEM(13)

           003 : 0810;  --         STORE   A       ;Store value in A, MEM(10)

           004 : 1404;  -- Here:   JUMP    Here    ;Loop here forever

           010 : 0000;  -- A:      DW      &H0000

           011 : 0004;  -- B:      DW      &H0004

           012 : 0003;  -- C:      DW      &H0003

           013 : 0003;  -- D:      DW      &H0003

END;

Appendix B: RISC scomp.vhd Implementation

LIBRARY IEEE;

LIBRARY LPM;

USE IEEE.STD_LOGIC_1164.ALL;

USE IEEE.STD_LOGIC_ARITH.ALL;

USE IEEE.STD_LOGIC_UNSIGNED.ALL;

USE LPM.LPM_COMPONENTS.ALL;

ENTITY SCOMP IS

  PORT(CLOCK,

       RESETN   : IN    STD_LOGIC;

       IO_ADDR  : OUT   STD_LOGIC_VECTOR(7 DOWNTO 0) );

END SCOMP;

ARCHITECTURE a OF SCOMP IS

  TYPE STATE_TYPE IS (RESET_PC, FETCH, DECODE, EX_LOAD, EX_STORE, EX_STORE2, EX_ADD, EX_JUMP,

                      EX_AND, EX_SHIFT, EX_ISTORE, EX_SUB, EX_JNEG, EX_JPOS, EX_JZERO, EX_OR, EX_XOR, EX_ADDI);

  SIGNAL STATE       : STATE_TYPE;

  SIGNAL AC,

         AC_SHIFTED,

         IR,

         MDR         : STD_LOGIC_VECTOR(15 DOWNTO 0);

  SIGNAL PC,

         MAR,

         MEM_ADDR    : STD_LOGIC_VECTOR(9 DOWNTO 0);

  SIGNAL MW          : STD_LOGIC;

  BEGIN

    -- Use LPM function to define unified program and data memory

    MEMORY: LPM_RAM_DQ

      GENERIC MAP (lpm_widthad         => 10,

                   lpm_outdata         => "UNREGISTERED",

                   lpm_indata          => "UNREGISTERED",

                   lpm_address_control => "UNREGISTERED",

                   lpm_file            => "TEST_CODE.mif",

                   lpm_width           => 16 )

      PORT MAP (data    => AC,

                q       => MDR,

                we      => MW,

                address => MEM_ADDR );

    -- Use LPM function to shift AC using the SHIFT instruction

    SHIFTER: LPM_CLSHIFT

      GENERIC MAP (lpm_width     => 16,

                   lpm_widthdist => 4,

                   lpm_shifttype => "LOGICAL" )

      PORT MAP (data      => AC,

                distance  => IR(3 DOWNTO 0),

                direction => IR(4),

                result    => AC_SHIFTED );

    MEM_ADDR  <= PC WHEN (STATE = FETCH) ELSE

                 MAR;                  -- Set memory address to PC when in fetch, MAR otherwise

    IO_ADDR   <= IR(7 DOWNTO 0);

    PROCESS (CLOCK, RESETN)

      BEGIN

        IF (RESETN = '0') THEN          -- Active low, asynchronous reset

          STATE <= RESET_PC;

        ELSIF (RISING_EDGE(CLOCK)) THEN

          CASE STATE IS

            WHEN RESET_PC =>

              MW        <= '0';        -- Clear memory write flag

              PC        <= "0000000000";  -- Reset PC to the beginning of memory, address 0x000

              AC        <= x"0000";    -- Clear AC register

              STATE     <= FETCH;      

            WHEN FETCH =>

              MW        <= '0';        -- Clear memory write flag

              IR        <= MDR;        -- Latch instruction into the IR

              MAR       <= MDR(9 DOWNTO 0);  -- MAR = least 10 bits of instruction (address)

              PC        <= PC + 1;     -- Increment PC to next instruction address

              STATE     <= DECODE;

            WHEN DECODE =>

              CASE IR(15 downto 10) IS

                WHEN "000000" =>       -- No Operation (NOP)

                  STATE <= FETCH;

                WHEN "000001" =>       -- LOAD

                  STATE <= EX_LOAD;

                WHEN "000010" =>       -- STORE

                  MW    <= '1';

                  STATE <= EX_STORE;

                WHEN "000011" =>       -- ADD

                  STATE <= EX_ADD;

                WHEN "000100" =>       -- SUB

                  STATE <= EX_SUB;

                WHEN "000110" =>       -- JNEG

                  STATE <= EX_JNEG;

                WHEN "000111" =>       -- JPOS

                  STATE <= EX_JPOS;

                WHEN "001000" =>       -- JZERO

                  STATE <= EX_JZERO;

                WHEN "001010" =>       -- OR

                  STATE <= EX_OR;

                WHEN "001011" =>       -- XOR

                  STATE <= EX_XOR;

                WHEN "001101" =>       -- ADDI

                  STATE <= EX_ADDI;

                WHEN "000101" =>       -- JUMP

                  STATE <= EX_JUMP;

                WHEN "001001" =>       -- AND

                  STATE <= EX_AND;

                WHEN "001100" =>       -- SHIFT (left or right)

                  STATE <= EX_SHIFT;

                WHEN "001110" =>       -- Indirect load (ILOAD)

                  MAR   <= MDR(9 DOWNTO 0);  -- Latch indirect address to the MAR

                  STATE <= EX_LOAD;

                WHEN "001111" =>       -- Indirect store (ISTORE)

                  MAR   <= MDR(9 DOWNTO 0);  -- Latch indirect address to the MAR

                  STATE <= EX_ISTORE;

                WHEN OTHERS =>

                  STATE <= FETCH;      -- Invalid opcodes default to NOP

              END CASE;

            WHEN EX_LOAD =>

              AC    <= MDR;            -- Latch data from MDR (memory contents) to AC

              STATE <= FETCH;

            WHEN EX_STORE =>

              MW    <= '0';            -- Drop MW,

              STATE <= EX_STORE2;      -- but do not transition to FETCH yet...

            WHEN EX_STORE2 =>          -- Second state ensures memory address

              STATE <= FETCH;          -- is stable while MW = 1

            WHEN EX_ADD =>

              AC    <= AC + MDR;       -- Add MDR (memory contents) to AC

              STATE <= FETCH;

            WHEN EX_SUB =>

              AC    <= AC - MDR;       -- Subtract MDR (memory contents) from AC

              STATE <= FETCH;

            WHEN EX_JNEG =>

              If ((AC /= "0000000000000000") AND(AC(15) = '1')) THEN 






PC <= IR(9 DOWNTO 0);       -- Jump if negative




  END IF;

              STATE <= FETCH;

            WHEN EX_JPOS =>

              If ((AC /= "0000000000000000") AND(AC(15) /= '1')) THEN 






PC <= IR(9 DOWNTO 0);       -- Jump if negative




  END IF;

              STATE <= FETCH;

            WHEN EX_JZERO =>

              If (AC = "0000000000000000") THEN 






PC <= IR(9 DOWNTO 0);       -- Jump if zero




  END IF;

              STATE <= FETCH;

            WHEN EX_OR =>




  AC <= AC OR MDR;       -- OR

              STATE <= FETCH;

            WHEN EX_XOR =>




  AC <= AC XOR MDR;       -- XOR

              STATE <= FETCH;

            WHEN EX_ADDI =>




  IF (IR(9) = '0') THEN




  AC <= IR(9) & "000000" & IR(8 DOWNTO 0) + AC;       -- ADDI for positive




  ELSE




  AC <= IR(9) & "111111" & IR(8 DOWNTO 0) + AC;       -- ADDI for negitive




  END IF;

              STATE <= FETCH;

            WHEN EX_JUMP =>            -- Set PC = lower 10 bits of IR

              PC    <= IR(9 DOWNTO 0);

              STATE <= FETCH;

            WHEN EX_AND =>

              AC    <= AC AND MDR;     -- AC = Bitwise AND of AC and MDR (memory contents)

              STATE <= FETCH;

            WHEN EX_SHIFT =>           -- AC = AC shifted 0-16 bits left or right

              AC    <= AC_SHIFTED;     --      Number of bits shifted and direction determined by least 5 bits of IR,

              STATE <= FETCH;          --        sign magnitude representation

            WHEN EX_ISTORE =>

              MW    <= '1';            -- MAR / MDR should now be stable, indicate write to memory

              STATE <= EX_STORE;       -- Proceed to the usual STORE state

            WHEN OTHERS =>

              STATE <= FETCH;          -- If an invalid state is reached, return to FETCH

          END CASE;

        END IF;

      END PROCESS;

  END a;
Appendix C: SCOMP GDF

Figure 1. Clock signal frequency.

Figure 2. MIF simulation waveform.

Figure C1. SCOMP GDF file showing clock connection.








